2.5　Mathematica による計算II

1　２つの２×３型行列の和を計算する．

$a=\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$ のように入力するには，行列パレットが必要であるが，ベクトルのように，リストとして入力するのが一般的である．行列名は大文字でなくてもよい．C,D,E,I,N,O は予約文字なので本来も目的以外には利用できない．

```
a={{1,2,3},{4,5,6}};
b={{0,1,2},{-1,2,4}};
a+b
```

答は　{{1,3,5},{3,7,10}}　である．

通常の括弧付きの行列として表示させるには，a+b の代わりに，MatrixForm[a+b] または a+b//MatrixForm とする．

2　行列の積を計算する．

a が 2×3 型でｂが 3×2 型の行列のとき積 ab を計算する．行列の積はドット(.)である．×や空白では計算できない．

```
a={{1,2,3},{4,5,6}};
b={{1,2},{3,4},{5,6}};
a.b
```

答は　{{22,28},{49,64}}　である．

3　連立一次方程式を解く．

$\begin{cases} x+2y+3z=4 \\ 2x+3y+4z=5 \end{cases}$ の解をもとめる．等号を重ねることと，未知数のリストを入力する．

```
Solve[{x+2y+3z==4,2x+3y+4z==5},{x,y,z}]
```

答は　{{x->-2+z, y->3-2z}} である．解の集合と読むと分かりやすい．

この問題は解が一意的でないので，答の前に，

Solve::svars:

方程式はすべての "solve" 変数に対しては解を与えない可能性があります．

というメッセージが現れる．

4　逆行列を求める．

文字成分の行列を扱う前に，ラベルを初期化しておく．必要なければ書かなくてよい．正則でない行列では，特異行列であるというエラーメッセージが出る．文字成分のときは，正則であるように扱われる．

```
Clear[a,b,c,d]
p={{a,b},{c,d}};
Inverse[p]
```

答は $\left\{\left\{\frac{d}{-b\,c+a\,d},-\frac{b}{-b\,c+a\,d}\right\},\left\{-\frac{c}{-b\,c+a\,d},\frac{a}{-b\,c+a\,d}\right\}\right\}$ である.

5　3次の単位行列を定義する.

```
IdentityMatrix[3]
```

が最も簡単である.

クロネッカーのデルタを自作するには mydelta[i_,j_]:=If[i==j,1,0] と関数定義する. mydelta[1,3] なら 0 で mydelta[2,2] なら 1 である.

```
mydelta[i_,j_]:=If[i==j,1,0]
Table[mydelta[i,j],{i,3},{j,3}]
```

6　2×3型の零行列を定義する. 3つの方法 z1,z2,z3 を述べる.

```
z1 = Table[Table[0, {3}], {2}]


z2 = Array[x, {2, 3}];
For[i = 1, i <= 2, i++,
   For[j = 1, j <= 2, j++, x[i, j] = 0]];
z2 // MatrixForm


<< LinearAlgebra`Master`
z3=ZeroMatrix[2, 3]
```

7　(2,3)行列とその転置行列を表示する.

```
u=Array[a,{2,3}];
v=Transpose[u];
u//MatrixForm
v//MatrixForm
```

8　3次の行列について結合法則を証明する. 最後に零行列になればよい. //Simplify を付けることで整理され確認できる.

```
p=Array[a,{3,3}]
q=Array[b,{3,3}]
r=Array[c,{3,3}]
(p.q).r-p.(q.r)//Simplify
```

9　係数行列または拡大係数行列に行基本変形を施し，上三角行列にする．

a={{1,2,3,4},{5,6,7,8},{9,10,11,12}} と入力する代わりに，Partition を使ってみる．Gauss の消去法は RowReduce により得られる．

```
a=Partition[{1,2,3,4,5,6,7,8,9,10,11,12},4]
RowReduce[a]//MatrixForm
```

答は $\begin{pmatrix} 1 & 0 & -1 & -2 \\ 0 & 1 & 2 & 3 \\ 0 & 0 & 0 & 0 \end{pmatrix}$ である．

10　行列の冪を計算する．

行列 a の n 乗は MatrixPower[a,n] である．n=-1 のときは逆行列を求める．

Exp[a] は a の各成分の指数を成分とする行列であり，MatrixExp[a] は a の指数行列を求める．

```
a = {{1.0, 1.0}, {1.0, 1.0}}
MatrixPower[a, 10]
Exp[a]
MatrixExp[a]
```

答は，{{512., 512.}, {512., 512.}}

{{2.71828, 2.71828}, {2.71828, 2.71828}}

{{4.19453, 3.19453}, {3.19453, 4.19453}}　である．

なお，a = {{1, 1}, {1, 1}} とすれば数値化されない結果をえる．

11　複素行列を計算する．

虚数単位 $i=\sqrt{-1}$ は Mathematica で I と表す．複素数 α の共役複素数は Conjugate[α] で得られる．Conjugate はリスタブル（リストに対して分配的）だから，行列 A の随伴行列 A^* は Transpose[Conjugate[A]] で求めることができる．あるいは，Conjugate[Transpose[A]] でもよい．次は，行列 A をエルミート行列と交代エルミート行列の和として表示する．

```
A = {{3, 2 + I}, {2 + 3I, 5 + 6I}};
B = Transpose[Conjugate[A]]
E1 = (A + B)/2;
E2 = (A - B)/2;
E1 + E2 // MatrixForm
Print[A // MatrixForm, "=", E1 // MatrixForm, "+", E2 // MatrixForm]
```

練習問題 2.5　次の問題を解決するためには，Mathematica にどのように入力すればよいか.

(1)　次の連立１次方程式を解くこと.

① $\begin{cases} 2x-y+5z+2w=1 \\ x+2y\quad+3w=-1 \\ 5y-5z-w=0 \\ -x-y-z+2w=3 \end{cases}$　② $\begin{cases} x_1+x_2+x_3-2x_4=0 \\ 2x_1+x_2-x_3+3x_4=0 \\ x_1-x_2+x_3-6x_4=0 \\ 3x_1+2x_2+x_3-2x_4=0 \end{cases}$

(2)　次の行列の逆行列を求めること.

① $\begin{pmatrix} 2 & -1 & 0 & 0 \\ -1 & 2 & -1 & -1 \\ 0 & -1 & 2 & 0 \\ 0 & -1 & 0 & 2 \end{pmatrix}$　② $\begin{pmatrix} 2 & -1 & 0 & 0 & 0 \\ -1 & 2 & -1 & 0 & 0 \\ 0 & -1 & 2 & -1 & 0 \\ 0 & 0 & -1 & 2 & -1 \\ 0 & 0 & 0 & -2 & 2 \end{pmatrix}$

(3)　次の行列に行基本変形を施し，上三角行列にせよ.

① $\begin{pmatrix} 2001 & 2002 & 2003 \\ 2004 & 2005 & 2006 \\ 2007 & 2008 & 2009 \end{pmatrix}$　② $\begin{pmatrix} 1! & 2! & 3! & 4! \\ 2! & 3! & 4! & 5! \\ 3! & 4! & 5! & 6! \\ 4! & 5! & 6! & 7! \end{pmatrix}$

(4)　３次の行列 a,b,c に対し [a,b]=ab-ba と定めるとき，次の等式を示すこと.

①　[a, bc]=[a,b]c+b[a,c]

②　[a,[b,c]]+[b,[c,a]]+[c,[a,b]]=0 （**ヤコビの恒等式**）